

Petlje

FOREACH petlja

Za svaki **član liste** koji se označava znakom ? primjenjuje naredbe koje su uglatoj zagradi

FOREACH :LISTA [naredbe ?]

Npr.

MAKE "L [1 2 3 4 5]

FOREACH :L [PR ?] ... ispis će biti brojevi 1 2 3 4 5

1. Procedura svaki broj iz liste množi s 2 ispisuje

TO BROJEVI

FOREACH [10 20 30 40 50] [PR ?*2]

End

Rezultat: -ispis- 20 40 60 80 100 (svaki broj u novom retku)

2. Procedura Crtanje za svaki član liste poziva proceduru PET :X i crta peterokut

TO CRTANJE

FOREACH [10 20 30 40 50] [PET ?]

END

TO PET :X

REPEAT 5 [FD :X RT 72]

END

Rezultat: Nacrtat će se 5 peterokuta duljine stranice 10, 20, 30, 40 i 50

3. Procedura SLOVO ispisuje prvi znak svakog člana liste

```
TO SLOVO  
FOREACH [ Marko Ivana Sanja Luka Ana Vedran ] [ PR FIRST ? ]  
END
```

Rezultat: ispis prvih slova iz svake riječi (M I S L A V – svako u novom retku)

4. Procedura SLOVA_NOVA iz liste :L dodaje prvi znak svakog člana u novu riječ NOVA i na kraju ispisuje riječ.

```
TO SLOVA_NOVA  
MAKE "L [ Marko Ivana Sanja Luka Ana Vedran ]  
MAKE "NOVA "  
FOREACH :L [ MAKE "NOVA WORD :NOVA FIRST ? ]  
PR :NOVA  
END
```

Rezultat: ispis nove riječi - MISLAV

5. Procedura SPIRALA crta spiralu uzimajući jedan po jedan znak iz liste

```
TO SPIRALA  
MAKE "L [ 10 20 30 40 50 60 70 80 90 100]  
FOREACH :L [ FD ? RT 90]  
END
```

6. Procedura ZBROJ zbraja sve članove liste

```
TO ZBROJ  
MAKE "B 0  
FOREACH [ 10 20 30 40 50 ] [ MAKE "B :B + ? ]  
PR :B  
END
```

Rezultat: ispis zbroja brojeva iz lista - 150

7. Procedura ZBROJ_SLOVA broji od koliko se znakova sastoji svaki član liste :L , dodaje taj broj u brojač i ispisuje rezultat.

```
TO ZBROJ_SLOVA
MAKE "L [ Marko Ivana Sanja Luka Ana Vedran ]
MAKE "B 0
FOREACH :L [ MAKE "B :B + COUNT ? ]
PR :B
END
```

Rezultat: 28

WHILE petlja

WHILE [logički uvjet] [naredbe]

Petlja započinje ključnom riječju **while** iza koje slijedi **logički uvjet** u uglatoj zagradi.

Naredbe unutar petlje (druga uglata zagrada) izvršavaju se sve dok je logički uvjet istinit

Ukoliko je logički uvjet lažan prije nego se petlja počne izvoditi, program će preskočiti čitavu petlju.

1. Procedura **ispis :x** ispisuje vrijednost varijable :x i povećava varijablu :x za 1 od početne (ulazne) vrijednosti, sve dok je :x manji od 10

```
to ispis :x
while [ :x < 10 ] [ pr :x   make "x :x + 1 ]
end
```

*Rezultat: Na poziv procedure **ispis 1** rezultat će biti brojevi od 1 do 9.*

2. Procedura **brojevi :x** povećava varijablu :x za 1 i ispisuje vrijednost varijable :x od početne (ulazne) vrijednosti, sve dok je :x manji od 10

```
to brojevi :x
while [ x < 10 ] [ make "x :x + 1   pr :x ]
end
```

*Rezultat: Na poziv procedure **brojevi 1** rezultat će biti brojevi od 2 do 10.*

3. Procedura CRTEZ :X crta peterokut duljine stranice :X. a procedura CRTANJE :X :P poziva proceduru CRTEZ :X i ponavlja crtanje dok je duljina stranice manja od 200, povećavajući stranicu za :P, što je ulazna vrijednost u proceduru.

```
to crtanje :x :p
while [ :x<200] [ setpc :x  crtez :x  make "x :x+:p]
end
```

```
to crtez :x
repeat 5 [ fd :x  rt 72 ]
end
```

Slučajni izbor **RANDOM**

RANDOM broj

RANDOM broj - slučajnim izborom bira broj od 0 do broj - 1

npr. RANDOM 10 će dati broj u rasponu 0 - 9

Primjer:

```
REPEAT 10 [ PR RANDOM 10 ]
```

Naredba će ispisati 10 brojeva izabranih slučajnim izborom

1. Procedura **sluc :x** ima ulaznu vrijednost **:x** (upisujemo jedan broj iz raspona 0 - 9), procedura slučajnim izborom bira broj iz raspona 0 - 9 i ispituje jesu li brojevi jednaki, ako jesu ispis je *pogodak*, ako nisu *pokušaj_ponovo*.

```
to sluc :x
make "s random 5
TEST :x = :s
IFT [ pr "pogodak ]
IFF [ pr "pokušaj_ponovo ]
pr :s
end
```

2. Procedura **LIK** slučajnim izborom bira boju i broj stranica lika i u odabranoj boji crta lik duljine stranice 100.

```
TO LIK
MAKE "B random 16
MAKE "N (random 6) + 3
SETPC :B
REPEAT :N [ FD 100 RT 360/:N ]
END
```

3. Procedura **BOJE** slučajnim izborom bira polumjer kružnice iz intervala 10-109, boju kružnice i boju ispune kružnice i crta odabrano.

```
to BOJE
make "x (random 100) + 10
setpc random 16
setfc random 16 circle :x fill
end
```